


★ネットで挑戦★

お絵かきツールを 作ろう

上田市マルチメディア情報センター／十勇士パソコンクラブ／PCN上田


1




画面に好きな絵がかける
お絵かきツールを作ろう

2

IchigoJam web



QRコード



IchigoJam webでも作れるよ！
fukuno.jig.jp/app/IchigoJam/

3


(1) 四角いカーソルを表示

```
10 CLV:CLS  
20 LC X,Y: ?CHR$(1);
```

2行のプログラムを入力します

4

入力できたら、F5で実行



5


(2)カーソルを点めつ

```
30 WAIT 3  
50 LC X,Y: ?CHR$(0);  
100 GOTO 20
```

3行のプログラムを追加します

6

入力できたら、F5で実行



7


(3)カーソルを左右に動かす

```
40 K=INKEY()  
60 X=X-(K=28)+(K=29)
```

2行追加します

8

入力できたら、F5で実行



9


(4)カーソルを上下にも動かす

```
70 Y=Y-(K=30)+(K=31)
```

1行追加します

10

入力できたら、F5で実行



上下の矢印キー↑↓で
カーソルが上下にも動く

11

(5) ドットをかく

2行を改造します

```
20 LC X,Y: ?CHR$( !SCR(X,Y) );  
50 LC X,Y: ?CHR$( !SCR(X,Y)^(K=32) );
```


改造

改造

「^」(ハット)
日本語キーボードでは「へ」
英語キーボードではShift+6

12

入力できたら、F5で実行



スペースキーを押すと
ドットがかける

- ドットがある所にカーソルを移動してもきえない
- ドットがある所でスペースキーを押すときえる
→ 消しゴムにもなる

13


(6) 画面クリアを追加

```
80 IF K=8 CLS
```

1行追加します

14

入力できたら、F5で実行



15

プログラムを保存

SAVE 0

ファイル0番がすでに使われている人は他の番号にしてください

16

(7) 絵を保存できるように

```
15 COPY #900, #D00, 768  
100 IF K!=10 GOTO 20  
110 COPY #D00, #900, 768:SAVE
```

3行追加します

17

入力できたら、F5で実行




18

保存された絵を呼び出すには

```
IchigoJam BASIC 1.4.2web jig.jp
OK
LOAD0
Loaded 228byte
OK
RUN
```

IchigoJamの電源ONの後
(IchigoJam webではページを開いた後)
LOAD0でプログラムを呼び出し
RUNで実行

19



前回かいた絵が呼び出される
※保存された絵をクリアしたい時は
BackSpaceキーで画面をクリアした後に
Enterキーで保存する

20

●お絵かきツール 完成

```
10 CLV:CLS 初期設定
15 COPY #900,#D00,768 前回の絵を呼び出し
20 LC X,Y: ?CHR$( !SCR(X,Y) ); カーソルを反転表示
30 WAIT 3 時間待ち
40 K=INKEY() キー入力          カーソルを反転消去
                                   スペースが押されたら消さない
50 LC X,Y: ?CHR$( !SCR(X,Y)^(K=32) );
60 X=X-(K=28)+(K=29) カーソルを左右に動かす
70 Y=Y-(K=30)+(K=31) カーソルを上下に動かす
80 IF K=8 CLS BackSpaceが押されたら画面クリア
100 IF K!=10 GOTO 20 Enterが押されていないならばもどる
110 COPY #D00,#900,768:SAVE 絵を保存
```


21

いろいろな絵をかいてみよう！


22

サンプル作品
「イチゴ太郎 vs コロナ星人」


23




24



25



26



27

絵の下に文字を表示して、紙しばいにする

```

10 CLV:CLS
15 COPY #900,#D00,768 } この2行は以前のまま。
                        } 描いた絵を画面に表示
20 LC 0,22   カーソルを画面下へ
30 ?"イチゴタロウ ハ キョウモ タノシク フロクラング
   グ。";   文章を表示
40 IF !INKEY() CONT   キーが押されるのを待つ
50 LRUN FILE()+1     次のファイルをLOADして実行

```

プログラムを改造すると、紙しばいになります。
何かキーを押すと、次へ進みます。

28

さらに改造しよう！

```
20 DRAW X,Y,2
```


```
50 DRAW X,Y,!POINT(X,Y)^(K=32)
```

2行を入力します

※IchigoJam バージョン1.4
またはIchigoJam webでできます

29

入力できたら、F5で実行



グラフィックで、2倍こまかい絵がかける

30